
Олімпіадні задачі і сортування

Дану тему, також варто розглядати з двох сторін. По-перше, при рішенні різних олімпіадних задач дані досить часто потрібно впорядкувати по деякій ознаці (тобто відсортувати). При цьому, якщо спеціально не оговорено інше, ми будемо вважати, що масив потрібно відсортувати в порядку неубування значень його елементів (для різних елементів — у порядку зростання). По-друге, задача сама по собі може вимагати побудови оптимального в контексті визначених вимог або нестандартного алгоритму сортування. Крім цього, ізюминка олімпіадної задачі може складатися у формалізації критерію, по якому варто сортувати дані.

Використання “стандартних” алгоритмів сортування одномірних масивів.

В усіх прикладах, що відносяться до одномірних масивів, будемо використовувати наступний тип даних: type list=array[1..N] of <скалярний тип>;

Конкретний тип елемента в більшості описуваних алгоритмів може бути як будь-яким числовим, так і символьним або навіть строковим. Останній тип скалярним не є, але до елементів цього типу застосовні всі операції порівняння, за допомогою яких звичайно (але не завжди !!!) і застосовується сортування.

Нехай у задачі дані необхідно відсортувати один раз. Тоді вибирати випливає той з відомих вам алгоритмів сортування, на реалізацію якого особисто у вас піде найменше часу і налагоджувати який теж не прийдеться. У такому випадку учасники звичайно реалізують так званий “бульбашкове” сортування або сортування прямим вибором. Звичайно, розмірність масиву при цьому повинна бути не більше 1000 елементів, у іншому випадку час роботи програми може виявитися невиправдано великим, тому що кількість тільки операцій порівняння в обох згаданих алгоритмах дорівнює N(N — 1)/2.

Якщо ж розмірність масиву велика або сортування потрібно проводити на кожнім кроці деякого циклу, то найпростіше використовувати алгоритм так називаної “швидкого” сортування [1-4]. Цей алгоритм на практиці виконується не більше Nlog2N операцій порівняння і найчастіше ще менше число операцій присвоювання, хоча теоретична оцінка для гіршого випадку є квадратичною по N. У мові програмування C він реалізований у виді функції qsort, що входить у бібліотеку stdlib. У повному постачанні середовища програмування Турбо Паскаль даний алгоритм можна знайти у файлі EXAMPLES\DOS\qsort.pas. Нерідко приведений у зазначеному файлі текст процедури сортування не потрібно змінювати взагалі, але іноді незначні переробки необхідні, тому корисно розібратися в особливостях даної реалізації алгоритму сортування. Розглянемо можливу модифікацію цієї процедури, виконану для задачі Кіровської відкритої командної олімпіади по програмуванню 2001 р. “Завбачливе журі”.

{задача}

У командних змаганнях з програмування часом для розв’язання задачі вважається час у хвилинах від початку туру до моменту посилки правильного рішення цієї задачі на перевірку. Журі намагається розташувати задачі в такому порядку, щоб для команди, що виконала один по одному всі задачі загальний час рішення був максимальним, якщо орієнтований час, що потрібно витратити на рішення кожної з задач відомо. Справді, якщо на тур пропонується дві задачі, і на рішення першої команда витрачає 10 хвилин, а на рішення другої— 20, то час рішення буде дорівнює 40 хвилинам (першу задачу команда здає на 10-й хвилині туру, другу — на 30-й, 10+30=40). Якщо ж задачі розташувати в зворотному порядку, то час буде дорівнювати 50 (задачі будуть здані на 20-й і 30-й хвилинах). Допоможіть членам журі розташувати задачі в бажаному порядку.

{\задача}

Очевидно, що задачі варто пропонувати в порядку незростання очікуваних тимчасових витрат на кожну з них. Тобто потрібно зробити сортування тимчасових характеристик наявних задач і переставити відповідно до отриманого порядку вихідні номери задач. Найпростіше робити це одночасно, незначно модифікуючи стандартну процедуру (зміни, внесені в реалізацію алгоритму сортування, виділені жирним шрифтом):

const nn=...{максимальна кількість задач};

type rr=record t,k:integer; end;

     list=array[1..nn] of rr;

var time:list;

    i,n:integer;

procedure QuickSort(var a: List; Lo, Hi: Integer);

procedure Sort(l, r: Integer);

var

  i, j, x: integer;

  y: rr;

begin

  i := l; j := r; x := a[(l+r) DIV 2].t;

  repeat

    while a[i].t > x do i := i + 1;

    while x > a[j].t do j := j - 1;

    if i <= j then

    begin

      y := a[i]; a[i] := a[j]; a[j] := y;

      i := i + 1; j := j - 1

    end

  until i > j;

  if l < j then Sort(l, j);

  if i < r then Sort(i, r)

end;

begin {QuickSort};

  Sort(Lo,Hi)

end;

begin {рішення задачі за допомогою модифікованої процедури QuickSort}

  read(n);

  for i:=1 to n do

  begin

    read(time[i].t);

    time[i].k:=i

  end;

  QuickSort(time,1,n);

{сортували часом, а друкуємо індекси}

  for i:=1 to n do

    write(time[i].k:8)

end.

На закінчення даного розділу приведемо таблицю, у якій для відомих універсальних алгоритмів сортування приведені порядки для кількості виконуваних тим або іншим алгоритмом в гіршому випадку операцій порівняння і присвоювання.

	Назва сортування
	Кількість порівнянь
	Кількість присвоювань

	Простий обмін (бульбашковий)
	O(N2)
	O(N2)

	Прямий вибір
	O(N2)
	O(N)

	Проста вставка
	O(N2)
	O(N2)

	Бінарна (подвійна) вставка
	O(NlogN)
	O(N2)

	Швидка (QuickSort)
	O(N2) (на практиці O(Nlog))
	O(N2) (на практиці O(Nlog))

	Злиттями
	O(NlogN)
	O(NlogN)

	Пірамідальна (HeapSort)
	O(NlogN)
	O(NlogN)


У такий спосіб найкращу теоретичну оцінку мають два останніх з перерахованих у таблиці алгоритмів, однак у практичному програмуванні для сортування даних звичайно використовується QuickSort, як у силу високої продуктивності (особливо виграє даний алгоритм по числу реально виконуваних присвоювань), так і в силу простої реалізації. Проте в олімпіадних задачах дані можуть бути представлені так, що відсортувати за відведений час їх можна буде лише за допомогою пірамідального сортування. Їх ще називають сортуванням купою або деревом [1-5]. Особливо корисним при рішенні деяких задач, навіть прямо із сортуванням не зв'язаних, може виявитися розуміння й уміння реалізувати операцію “проштовхування” елемента по упорядкованому дереву, за допомогою якої і здійснюється дане сортування. Додаткову інформацію про різні алгоритми сортування можна одержати в [6].

Оптимальне сортування

Нехай нам потрібно побудувати алгоритм сортування, оптимальний по кількості виконуваних при цьому операцій присвоювання або порівняння для гіршого випадку. З приведеної вище таблиці випливає, що алгоритм прямого вибору є лінійним по кількості виконуваних операцій присвоювання (їхня кількість для гіршого випадку дорівнює 3(N — 1)) і, отже, асимптотично оптимальний. Більш того, при використанні додаткового масиву для запису результату цей алгоритм можна реалізувати рівно за N переміщень елементів, що є точною нижньою оцінкою для кількості присвоювань у гіршому випадку, а саме — якщо всі N елементів спочатку знаходилися не на своїх місцях, ми не можемо менш, ніж за N переміщень, розставити їх у потрібному порядку. Дану властивість алгоритму прямого вибору можна використовувати і у практичному програмуванні, коли при сортуванні деяких об'єктів операція присвоювання (переміщення) виявляється значно більш трудомісткою, ніж операція порівняння. Наприклад, при сортуванні стовпців двомірного масиву відповідно до значень перших елементів стовпців або при сортуванні записів, за значенням одного з полів, або при сортуванні дійсних контейнерів, що потрібно переставити за допомогою піднімального крана по зростанню вартості вантажів, що знаходяться в них. Справедливості заради помітимо, що таку технічну проблему при комп'ютерному сортуванні можна вирішувати і по-іншому: сортувати можна не самі об'єкти, а покажчики на них.

На олімпіаді задача побудови оптимального по кількості переміщень алгоритму сортування може бути поставлена й інакше. Наприклад, спробуйте знайти оптимальне рішення задачі “Паркування”, що пропонувалося на міжнародній олімпіаді по інформатиці в 2000 році. Врахуйте, що для кожної з конкретних вхідних неупорядкованих послідовностей, воно може бути іншим. Визначте чи для будь-яких значень N з умови задачі це можна зробити в загальному випадку. Універсальне рішення цієї задачі, розглянуте в [7], є евристичним і не завжди мінімально, хоча і задовольняє умові.

Перейдемо тепер до розгляду алгоритму сортування, оптимального по кількості порівнянь елементів між собою. Якщо споконвічно нам не відомі ніякі відносини між елементами, то для N елементів результатом їх сортування може виявитися кожна з N! перестановок елементів. Тоді з теорії інформації випливає, що нижня границя числа порівнянь, необхідних у гіршому випадку для визначення потрібної нам перестановки, дорівнює log2N!. При великих N це число росте так само, як і Nlog2N — теоретична оцінка зверху для ряду універсальних алгоритмів (див. таблицю вище). Наприклад, алгоритм вставки на першому кроці порівнює елементи a1 і a2, на другому — в упорядковану послідовність з 2-х елементів вставляється елемент a3 і т.д. Якщо пошук місця для вставки чергового елемента у вже упорядковану послідовність зробити бінарним, то число порівнянь у гіршому випадку складе log22 + log23 + … + log2N і буде відрізнятися від log2N!, менш ніж на N. Складемо таблицю, у якій для невеликих N приведені порівняльні значення результатів обчислення по трьох формулах:

	N
	2
	3
	4
	5
	6
	7
	8
	9
	10
	11
	12

	log2N!
	1
	3
	5
	7
	10
	13
	16
	19
	22
	26
	29

	бін.вставка
	1
	3
	5
	8
	11
	14
	17
	21
	25
	29
	33


	Nlog2N
	2
	6
	8
	15
	18
	21
	24
	36
	40
	44
	48


Нескладно показати, що й інші універсальні алгоритми сортування, що мають ту ж асимптотичну оцінку на кількість порівнянь, не збігаються по кількості порівнянь з точною нижньою оцінкою. У задачниках по програмуванню зустрічається вправа, що вимагає відсортувати 5 довільних елементів не більш, ніж за 7 порівнянь (переконаєтеся, що будь-який універсальний алгоритм зробить це, у гіршому випадку використовуючи 8 порівнянь, і спробуйте самостійно побудувати алгоритм для рішення цієї задачі, тому що це дійсно можливо зробити за 7 порівнянь). 

У [8] показано, як для відомого фіксованого значення N за абсолютно мінімальне число порівнянь відсортувати конкретну вхідну послідовність з N елементів. Для цього можна використовувати так званий метод центрів ваг. Нехай деяка кількість порівнянь уже була проведена, тобто нашу безліч елементів частково впорядковано. Назвемо лінійними продовженнями такої безлічі ті перестановки з N елементів, що не суперечать уже наявному частковому упорядкуванню. Тоді a(i) – середнє місце i-го елемента з вхідного масиву a в упорядкованому масиві, по наявній на дійсний момент інформації можна підрахувати так:

[image: image1.wmf].

)

(

)

(

1

)

(

å

=

я

продолжени

линейные

i

p

a

k

i

a

Тут p(i) — місце i-го елемента в лінійному продовженні, k(a) - загальне число лінійних продовжень елементів масиву a. Два різних елементи i і j не порівнянні тоді і тільки тоді, коли b(i, j) = |a(i) — a(j)| < 1. Більш того, у черговому порівнянні повинні брати участь елементи i  і j, значення b(i, j) у яких мінімально й i < j. Після цього кількість лінійних продовжень зменшується, середні місця елементів у них перераховуються і вибирається нова пара елементів для порівняння. Відсортований масив відповідає єдиному лінійному продовженню, що залишилося, місця всіх елементів у якому визначені, і b(,, j) ( 1, якщо ( ( j.

Покажемо роботу цього алгоритму на наступному прикладі. Нехай для 5 елементів уже відомо, що a1 ( a2 ( a3 і a4 ( a5. Перелічимо всі 10 лінійних продовжень цієї безлічі і підрахуємо середнє місце кожного з 5 елементів у них:

	4
	5
	1
	2
	3
	a(1)=(6(1+3(2+1(3)/10=1,5

	4
	1
	5
	2
	3
	a(2)=(3(2+4(3+3(4)/10=3

	4
	1
	2
	5
	3
	a(3)=(1(3+3(4+6(5)/10=4,5

	4
	1
	2
	3
	5
	a(4)=(4(1+3(2+2(3+1(4)/10=2

	1
	4
	5
	2
	3
	a(5)=(1(2+2(3+3(4+4(5)/10=4

	1
	4
	2
	3
	5
	Мінімальні b(i, j):

	1
	4
	2
	3
	5
	b(1,4)=b(3,5)=0,5

	1
	2
	4
	5
	3
	Отже, порівнюємо, наприклад, 1-й і 4-й елементи, забираємо

	1
	2
	4
	3
	5
	 невірними лінійні продовження, що стали, і знову перераховуємо

	1
	2
	3
	4
	5
	a(i) і т.д.


Тепер можна показати в якому порядку оптимальний по кількості порівнянь алгоритм повинний порівнювати елементи споконвічно ніяк не упорядкованого масиву. До початку сортування припустимими лінійними продовженнями є всі N! перестановок елементів масиву, всі елементи  якого рівноправні. Так, для 5 елементів усі a(i) = 3, а b(i, j) = 0. Тоді спочатку ми порівнюємо два будь-яких елементи, а потім — два будь-яких із що залишилися, тому що значення b для них після першого порівняння залишиться нульовим, тобто мінімально можливим. Подібним чином, виконуються порівняння на першому кроці, наприклад, алгоритму сортування злиттям. Нехай тепер відомо, що a1 ( a2 і a4 ( a3. Очевидно, що b(1, 4) = b(2, 3) = 0, причому самі значення a для згаданих елементів можна і не підраховувати. Таким чином, на третьому кроці ми повинні порівняти, наприклад, максимальні елементи в упорядкованих парах. Нехай виявиться, що a2 ( a3. Тоді ті що залишились 15 лінійним продовженням легко підрахувати значення a: a(1) = 1,6; a(2) = 3,2; a(3) = 4,8; a(4) = 2,4; a(5) = 3. Тобто на четвертому кроці, відмінному від універсальних сортувань, порівнянню підлягають елементи a2 і a5. Поза залежністю від результатів цього порівняння за 3 порівняння, що залишилися масив буде упорядкований.

Відкритим залишається питання, чи завжди кількість порівнянь у такому оптимальному алгоритмі збігається з точною нижньою оцінкою для всіх алгоритмів сортування. На жаль, це не так. Уже при N = 12 оптимальному алгоритмові буде потрібно зробити 30 порівнянь замість 29 очікуваних. Подальше вивчення цього питання утруднено в силу надзвичайної трудомісткості алгоритму, заснованого на методі центрів ваг.

На закінчення даного розділу покажемо, як може бути сформульована олімпіадна задача, у рамках якої потрібно побудувати алгоритм оптимальний або асимптотично оптимальний по кількості порівнянь. 

{задача}

Вхідним параметром у задачі є число N. Після його зчитування ваша програма повинна видавати номера елементів, порівняти які їй потрібно на черговому кроці, і зчитувати результат цього порівняння. Коли буде зроблене достатнє для вашого алгоритму число порівнянь, програма повинна видати індекси елементів у масиві в порядку неубування значень відповідних елементів. 

{\задача}

Замість діалогового режиму роботи може бути запропонована спеціальна бібліотечна функція зі створеного спеціально для цієї задачі модуля, шляхом звертання до якої можна буде одержувати інформацію про співвідношення значень у пари елементів.

Сортування даних спеціального виду

Вище було показано, як варто використовувати універсальні алгоритми сортування, придатні для будь-якого виду даному, підлягаючому порівнянню. Однак, найчастіше інформація про безліч значень, що можуть приймати елементи масиву, може змінити підхід до сортування. Нехай, наприклад, потрібно відсортувати 100000 цифр, що вводяться, наприклад, зі стандартного потоку введення (це типова задача для районних або міських олімпіад по інформатиці). Як уже говорилося в лекції 3, максимальний розмір масиву навіть однобайтових елементів не може перевищувати 64 кілобайта. Крім того, час роботи кожного з універсальних “ефективних” (тобто виконуючого порядку Nlog2N дій) для 100000 елементів швидше за все буде перевищувати допустиме за умовою час рішення задачі. У цьому випадку алгоритм сортування фактично полягає в підрахунку кількості кожної з цифр і запису результату відповідно до отриманих результатів підрахунку. Реалізувати цей алгоритм можна так:

var d:array[0..9] of longint; {масив для підрахунку}

    a:byte;{змінна для введення цифр}

    n,i,j:longint;

begin

  read(n);{n - кількість цифр}

  fillchar(d,sizeof(d),0);

  for i := 1 to n do
  begin

    read(a);

    d[a] := d[a] + 1

  end;

  for j := 0 to 9 do {друкуємо результат}

    for i := 1 to d[j] do
      write(j:2)

end.

Помітимо, що ця програма працює вірно й у випадку, коли якоїсь з цифр у послідовності, що вводиться, немає зовсім. Подібний алгоритм називають сортуванням підрахунком або “кишеньковим” сортуванням [1, 5]. Його трудомісткість складає O(N+m) (а саме, 2N + m операцій), де m — кількість різних елементів у масиві. Очевидно, що якщо m(N і ми можемо відвести пам'ять для підрахунку кількості кожного з m можливих значень для елементів масиву, то алгоритм виявиться лінійним відносно N. Якщо ж m>N, то алгоритм може як вигравати, так і програвати в порівнянні з універсальними ефективними алгоритмами. Наприклад, для масиву з 1000 позитивних чисел типу integer сортування підрахунком виконується за 32767*2 + 1000 = 66534 операції, а, наприклад, сортування злиттям, — за 2*1000*log21000=20000 операцій. Але вже для 10000 таких же чисел сортування підрахунком виконується за 75534 операції, а злиттям — за 280000.

Як вправу, що ілюструє ту ж саму ідею, що і сортування підрахунком, реалізуйте програму, що визначає, яка буква зустрічається в тексті найчастіше. Нагадуємо, що в Турбо Паскалі індексами масиву можуть служити і символи, у даному випадку це можна використовувати при описі додаткового масиву. Не забудьте також, що одній і тій же букві алфавіту відповідають 2 різних символи, що варто врахувати при підрахунку (а для російських букв це не настільки просто як для латинських).

Ідею сортування підрахунком продовжує “цифрове” сортування, об'єктами якого можуть служити довільні числа і навіть рядок. Такий алгоритм раніше використовувався для сортування перфокарт [5]. У перфокартах цифри кодувалися одиночними дірками в рядках 0-9 відповідного стовпчика. Сортувальній машині вказували стовпець для сортування і вона розкладала перфокарти на 10 стопок у залежності від того, яка з дірок була пробита. Як же за допомогою цієї машини можна було відсортувати колоду перфокарт із багатозначними цифрами? Як не дивно, процедуру сортування починали не зі старшого розряду, а з молодшого. Отримані в результаті першого сортування 10 стопок знову складали в одну колоду (починаючи з нулів у молодшому розряді і закінчуючи дев'ятками) і сортували вже по розряду десятків і т.д. Якщо числа були k-значними, то після k операцій порозрядного сортування колода виявлялося упорядкованою. Продемонструємо це на прикладі тризначних чисел (кожен стовпець, починаючи з другого, показує результат сортування вихідного стовпця по черговому розряді):

	329
	720
	720
	329

	457
	355
	329
	355

	657
	436
	436
	436

	839
	457
	839
	457

	436
	657
	355
	657

	720
	329
	457
	720

	355
	839
	657
	839


Коли числа сортуються по якому-небудь розряді важливо, щоб застосовуваний при цьому алгоритм сортування був стійким, тобто числа, у яких в поточному розряді - ті самі цифри, зберігали те ж взаємне розташування, яке було між ними перед сортуванням по цьому розряді. Стійким є, наприклад, модифікований алгоритм сортування підрахунком, у якому в елементі допоміжного масиву d[х] буде зберігатися кількість елементів у масиві, не переважаючих х. Покажемо, як зміниться при цьому програма сортування масиву цифр, розташованих у змінній a типу list (результат помістимо в масив b того ж типу):

  fillchar(d,sizeof(d),0);

  for i := 1 to n do
    d[a[i]] := d[a[i]] + 1;

  for j := 1 to 9 do 

    d[j] := d[j] + d[j-1];

{d[j] – кількість елементів не переважаючих j}

  for i := n downto 1 do

    begin

      b[d[a[i]]] := a[i];

      d[a[i]] := d[a[i]]-1

    end
Справді, у відсортованому масиві a[і] свідомо може стояти на місці d[a[і]], адже саме стільки елементів у масиві не перевершують a[і]. Потім d[a[i]] зменшується на одиницю і інші елементи масиву, рівні a[i], будуть записані лівіше. Взаємний порядок між рівними елементами при цьому не порушиться, тому що при записі результату масив проглядається з кінця.

Оцінимо час роботи алгоритму цифрового сортування, з порозрядним використанням сортування підрахунком. Для N чисел з k знаками (або для рядків довжини k), у записі яких беруть участь m різних чисел (симоволів), кількість операцій має порядок O(k + km). Якщо k фіксовано і m(N, то загальна кількість дій має порядок O(N). Як і раніше коефіцієнт у такому лінійному алгоритмі варто порівнювати зі значенням log2N, для визначення області його ефективного застосування. Однак цифрове сортування зовсім незамінне для упорядкування масивів даних, що мають кілька різних полів. Наприклад, при сортуванні дат, заданих значенням року, місяця і числа і т.п.

На ідеї, схожої з цифровим сортуванням, засноване рішення задачі “Редактор” I Всеросійської командної олімпіади по програмуванню [9].

 Задачі зворотні сортуванню 

Нехай задана конкретна реалізація алгоритму сортування і значення N. Потрібно відновити деякі його характеристики. Подібна задача пропонувалася на відкритій командній олімпіаді по програмуванню в Санкт-Петербурзі в 1996 році.

{задача}

Нижче приведений алгоритм MSort, що сортує злиттям заданий масив A з N цілих чисел. 

Потрібно написати програму, що для заданого N будує приклад масиву A, на якому досягається максимальне число порівнянь. Всі елементи масиву A повинні бути різними і знаходитися в діапазоні від 1 до N.

	Приклад вихідних даних:
	Результат:

	3
	2 1 3


procedure MSort (N: integer; var A: list);

var Help: list;

  procedure Make (u, v: integer);

  var i, j, m, k: integer;

  begin

    if u < v then
     begin

        m := (u+v) div 2;

        Make (u, m);

        Make (m+1, v);

        i := u; j := m+1; k := u;

        while (i<=m) and (j<=v) do
        begin

          if A [i] < A [j] then begin Help[k] := A[i]; i := i+1 end
                           else begin Help[k] := A[j]; j := j+1 end;

           k := k+1

        end;

        while i<=m do begin Help[k] := A[i]; i:=i+1; k:=k+1 end;

        while j<=v do begin Help[k] := A[j]; j:=j+1; k:=k+1 end;

        for i := u to v do A[i] := Help[i]

      end

   end

begin {MSort}

   Make (1, N)

end;

{\задача}

Приведемо рекурсивну процедуру, що вирішує цю задачу:

procedure Fill (var A: list; u, v: integer; start, step: integer);

begin

  if u = v then A [u] := start

  else begin

    Fill (A, u, (u+v) div 2, start, step*2);

    Fill (A, (u+v) div 2 + 1, v, start + step, step*2)

  end

end;

Масив А буде заповнений у результаті наступного звертання до такої процедури: Fill(A, 1, N, 1, 1). Крім того, можна задати визначити кількість порівнянь, виконуваних алгоритмом у гіршому випадку, причому в такій задачі обмеження на величину N уже практично відсутні, тому вирішити її шляхом підстановки заповненого масиву A у вихідну процедуру сортування і включення в неї лічильника порівнянь неможливо. Спробуйте написати програму, що підраховує кількість порівнянь, виконуваних приведеною вище процедурою сортування масиву злиттям, за введеним значенням N. Саму процедуру сортування, а також рекурсію, у програмі використовувати не потрібно. Подібні задачі можна поставити і для інших алгоритмів сортування. Побудуйте приклад масиву, на якому стандартна процедура QuickSort буде виконувати максимальну кількість порівнянь. Ця задача має і практичне значення. По побудованим для різних значень N прикладам можна зрозуміти для яких саме масивів швидке сортування не є ефективною і чи дійсно в цьому випадку кількість порівнянь має порядок O(N2).

